



A Tutorial Introduction to Maxima

Nidish Narayanaa B

Department of Aerospace Engineering

Indian Institute of Space Science & Technology, Trivandrum

FOSS Group, IIST, 2017



Outline

Introduction

Basics

Plotting

Solving Equations

Vectors and Matrices

- Vectors
- Matrices and Linear Algebra

Calculus

- Differential Calculus
- Taylor Series
- Symbolic Integration

Symbolic Manipulations



What is Maxima?

*(From the website) Maxima is a system for the **manipulation of symbolic and numerical expressions**, including differentiation, integration, Taylor series, Laplace transforms, ordinary differential equations, systems of linear equations, polynomials, sets, lists, vectors, matrices and tensors.*



What is Maxima?

- ▶ Powerful Computer Algebra System (CAS) combining symbolic, numerical, and graphical capabilities



What is Maxima?

- ▶ Powerful Computer Algebra System (CAS) combining symbolic, numerical, and graphical capabilities
- ▶ Cousin of the commercial Macsyma CAS (currently available without support)



What is Maxima?

- ▶ Powerful Computer Algebra System (CAS) combining symbolic, numerical, and graphical capabilities
- ▶ Cousin of the commercial Macsyma CAS (currently available without support)
- ▶ Completely Free and Open Source Software, written in LISP



What is Maxima?

- ▶ Powerful Computer Algebra System (CAS) combining symbolic, numerical, and graphical capabilities
- ▶ Cousin of the commercial Macsyma CAS (currently available without support)
- ▶ Completely Free and Open Source Software, written in LISP
- ▶ Fully customizable and extensible



Interface

maxima Native CLI

```

maxima 80x24
~ maxima
Maxima 5.37.2 http://maxima.sourceforge.net
using Lisp GNU Common Lisp (GCL) GCL 2.6.12
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
The function bug_report() provides bug reporting information.
(%i1) A:matrix([1,2,3],[2,3,1],[4,3,5]);
                                [ 1 2 3 ]
                                [ 2 3 1 ]
(%o1)                            [ 4 3 5 ]
                                [  ]
                                [  ]
                                [  ]
(%i2) gamma:alpha+beta;
(%o2)                            beta + alpha
(%i3)

```




Interface

maxima Native CLI
xmaxima History :)

```
File Edit Options Maxima Help
[Maxima Icon]
(%i1) A:matrix([1,2,3],[2,3,1],[4,3,2]);
                                     [ 1 2 3 ]
(%o1)                                [ 2 3 1 ]
                                     [ 4 3 2 ]
(%i2) gamma:alpha+beta;
(%o2)                                beta + alpha
(%i3) |
```

Started Maxima



Interface

maxima Native CLI

xmaxima History :)

wxmaxima Wxwidgets

```
(%i2) A:matrix([1,2,3],[2,3,1],[3,2,4]);
      gamma:alpha+beta;
      (%o1) 
$$\begin{bmatrix} 1 & 2 & 3 \\ 2 & 3 & 1 \\ 3 & 2 & 4 \end{bmatrix}$$

      (%o2)  $\beta + \alpha$ 
```

Welcome to wxMaxima

Ready for user input



Interface

maxima Native CLI
 xmaxima History :)
 wxmaxima Wxwidgets
 imaxima Emacs!

```
File Edit Options Buffers Tools Maxima YASnippet Complete In/Out Signals Help
Maxima 5.37.2 http://maxima.sourceforge.net
using Lisp GNU Common Lisp (GCL) GCL 2.6.12
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
The function bug-report() provides bug reporting information.
(N1) block(load("/usr/share/maxima/5.37.2/enacs/lnmaxima.lisp")), linenum:0)
(N1) A1matrix([1,2,3],[3,2,1],[4,2,3]);

      (N1)                                     
$$\begin{pmatrix} 1 & 2 & 3 \\ 3 & 2 & 1 \\ 4 & 2 & 3 \end{pmatrix}$$


(N2) gamma:alpha+beta;
      (N2)                                     
$$\beta + \alpha$$

(N3) █
```

U:0** *maxima* 1.15 (Inferior Maxima: run Golden VM! ug company yas helm) [53.6%]7:09PM 0.25



Interface

maxima Native CLI

xmaxima History :)

wxmaxima Wxwidgets

imaxima Emacs!

```
(%i2) A:matrix([1,2,3],[2,3,1],[3,2,4]);
      gamma:alpha+beta;
      (%o1) 
$$\begin{bmatrix} 1 & 2 & 3 \\ 2 & 3 & 1 \\ 3 & 2 & 4 \end{bmatrix}$$

      (%o2)  $\beta + \alpha$ 
```

Welcome to wxMaxima

Ready for user input



Maxima Init File

- ▶ By default, the variable “maxima_userdir” stores the user directory (in Linux it is `/.maxima`)
- ▶ You will have to create this directory and a file “maxima-init.mac” inside it
- ▶ Try adding “disp(“Greetings!”)\$” in the file and fire up maxima

Command Summary



`disp` display string to stdout

`describe,?` describe maxima command or variable

`apropos,??` inexact search for maxima command or variable



Outline

Introduction

Basics

Plotting

Solving Equations

Vectors and Matrices

- Vectors
- Matrices and Linear Algebra

Calculus

- Differential Calculus
- Taylor Series
- Symbolic Integration

Symbolic Manipulations



Some basics

- ▶ Statements terminated by a “;”
- ▶ Suppress output by terminating with a \$
- ▶ Variable assignment with “:”
- ▶ Function declaration with “:=”
- ▶ **Note:** “=” is a symbol, not an operator!
- ▶ Lists indices go as 1,2,3,...
- ▶ “nouns” and “verbs”



Some basics

Maxima

```
diff(sin(x^2),x);
```

$$2x \cos x^2$$

Maxima

```
integrate(cos(x)*cosh(x),x);
```

$$\frac{e^{-x} ((e^{2x} - 1) \cos x + (e^{2x} + 1) \sin x)}{4}$$



Basic Arithmetic

Handling literals

Maxima

```
(1/2+sqrt(2/3))^2;
```

Maxima

```
expand((1/2+sqrt(2/3))^2);
```

$$\left(\frac{1}{2} + \frac{\sqrt{2}}{\sqrt{3}}\right)^2$$

$$\frac{\sqrt{2}}{\sqrt{3}} + \frac{11}{12}$$



Basic Arithmetic

Handling literals

Maxima

```
expand((1/2+sqrt(2/3))^2),  
numer;
```

1.733163247594392

Maxima

```
fpprec:20;  
bfloat(expand(  
  (1/2+sqrt(2/3))^2));
```

1.7331632475943926994B0



Algebra

Handling Symbolics

Maxima

```
expand((x+3*y+x^2*y)^3);
```

Maxima

```
(x+3*y+x^2*y)^3;
```

$$(x + 3y + x^2 y)^3$$

$$\begin{aligned} & x^6 y^3 + 9 x^4 y^3 + 27 x^2 y^3 \\ & + 27 y^3 + 3 x^5 y^2 \\ & + 18 x^3 y^2 + 27 x y^2 \\ & + 3 x^4 y + 9 x^2 y + x^3 \end{aligned}$$

Algebra



Handling Symbolics

Maxima

```
expand(subst([x=5/y],
(x+3*y+x^2*y)^3));
```

$$27y^3 + 810y + \frac{8100}{y} + \frac{27000}{y^3}$$

Maxima

```
factor(27*y^3+810*y+8100/y
+2700/y^3);
```

$$\frac{27(100 + 300y^2 + 30y^4 + y^6)}{y^3}$$

Algebra



Handling Symbolics

Maxima

```
solve([expand(subst([x=5/y],(x+3*y+x^2*y)^3))=0],y);
```

$$[y = -\sqrt{10}i, y = \sqrt{10}i]$$

Algebra



Handling Symbolics

Maxima

```
solve([x+y+z=4,y+z-2*x=5,z=4*x-3*y+t,t+y=z],[x,y,z,t]);
```

$$\left[\left[x = -\frac{1}{3}, y = -\frac{1}{3}, z = \frac{14}{3}, t = 5 \right] \right]$$

Algebra



Handling Symbolics

Maxima

```
sum(1/n^2,n,0,N);
```

$$\sum_{n=0}^N \frac{1}{n^2}$$

Maxima

```
product(1/n^2,n,1,10);
```

$$\frac{1}{13168189440000}$$



Commands Summary

`diff` symbolic differentiation

`integrate` symbolic integrations

`numer` numerical result

`fpprec` (variable) set floating point precision

`bfloat` floating point representation

`expand` expand expression

`factor` factor expression

`subst` substitute for variables

`solve` solve system



Outline

Introduction

Basics

Plotting

Solving Equations

Vectors and Matrices

- Vectors
- Matrices and Linear Algebra

Calculus

- Differential Calculus
- Taylor Series
- Symbolic Integration

Symbolic Manipulations





Plotting functions

Explicit function plotting

Maxima

```
plot2d(sin(x), [x, -\%pi, \%pi]);  
plot3d(sin(x+y)*cos(x-y), [x, -\%pi, \%pi], [y, -\%pi, \%pi]);
```

pointer: try adding “wx” as a prefix in wxmaxima






Plotting functions

Implicit plots

Maxima

```
load(implicit_plot);  
implicit_plot(x^2=y^2-3*y+1,[x,-4,4],[y,-4,4],  
[gnuplot_preamble,"set\ zeroaxis"]);
```






Plotting functions

parametric plots

Maxima

```
r(t):=t;  
plot2d([parametric,r(t)*cos(t),r(t)*sin(t),  
[t,-2*%pi,2*%pi],[nticks,80]], [x,-7,7]);
```






Plotting functions

Maxima

```
expr_1:cos(y)*(12.0+6*cos(x));  
expr_2:sin(y)*(12.0+6*cos(x));  
expr_3:6*sin(x);  
plot3d([expr_1,expr_2,expr_3],  
[x,0,2*%pi],[y,0,2*%pi],[gnuplot_pm3d,true]);
```

Parametric function






Plotting functions

Misc

Maxima

```
set_plot_option([gnuplot_preamble,"set\ cntrparam  
levels\ 12"]);  
contour_plot(sin(y)*cos(x)^2,[x,-4,4],[y,-4,4]);
```





Plotting functions


Discontinuous function

Maxima

```
g(x):=\ if\ x>=1\ and\ x<2\ then\ (x-1)\ elseif\ x>=2
and\ x<=3\ then\ (6-2*x)\ else\ 0;
```

$$g(x) := \text{if } x \geq 1 \wedge (x < 2) \text{ then } x - 1 \text{ elseif } x \geq 2 \wedge (x \leq 3) \text{ then } 6 - 2x \text{ else } 0$$

Plotting functions



Commands Summary



`plot2d` 2d plots

`plot3d` 3d plots

`implicit_plot` plotting implicit expressions

`contour_plot` contours



Outline

Introduction

Basics

Plotting

Solving Equations

Vectors and Matrices

- Vectors
- Matrices and Linear Algebra

Calculus

- Differential Calculus
- Taylor Series
- Symbolic Integration

Symbolic Manipulations



Systems of equations

Maxima

```
apropos("solve");
```

```
[baksolve, desolve, funcsolve, globalsolve, linsolve,  
 linsolvewarn, linsolve_by_lu, linsolve_params,  
 solve, solvedecomposes, solveexplicit, solvefactors,  
 solvenullwarn, solveradcan, solvetrigwarn, tmlinsolve]
```



Systems of equations

linear systems

Maxima

```
eq1:x+y=8*a*x-z+4*5;\ eq2:y+z*\%pi=4*x;
eq3:x-y+z=4;
linsolve([eq1,eq2,eq3],[x,y,z]);
```

$$\left[\begin{aligned} x &= -\frac{8 + 12\pi}{(4 + 4\pi)a - \pi - 4}, y = -\frac{32 + 8\pi + 16\pi a}{(4 + 4\pi)a - \pi - 4}, \\ z &= \frac{16a - 40}{(4 + 4\pi)a - \pi - 4} \end{aligned} \right]$$



Systems of equations

Algebraic Equation (analytic solution)

Maxima

```
solve(a*x^2+b*x+c=0,x);
```

$$\left[x = -\frac{b + \sqrt{b^2 - 4ac}}{2a}, x = \frac{\sqrt{b^2 - 4ac} - b}{2a} \right]$$



Systems of equations

Substitute values

Maxima

```
sol:solve(a*x^2+b*x+c=0,x);  
radcan(subst([a=1,b=2,c=3],sol[1]));
```

$$x = -\sqrt{2}i - 1$$



Systems of equations

System of algebraic equations

Maxima

```
f(x):=x^2-1;
solve([f(x/y)=0,f(x*y)=0],[x,y]);
```

$$f(x) := x^2 - 1$$

$$[[x = -1, y = 1], [x = 1, y = -1], [x = -i, y = -i], [x = i, y = i], [x = -i, y = i], [x = i, y = -i], [x = -1, y = -1], [x = 1, y = 1]]$$



Systems of equations

Polynomials & Numerical solutions

Maxima

```
allroots(x^5+x^2-x+8=0);
```

$$\begin{aligned} &[x = 0.9189543576112322 i + 1.179427901164746, \\ & \quad x = 1.179427901164746 - 0.9189543576112322 i, \\ & \quad x = 1.427958893635229 i - 0.3522627040736663, \\ & \quad x = -1.427958893635229 i - 0.3522627040736663, \\ & \quad x = -1.65433039418216] \end{aligned}$$



Systems of equations

Checking Solution

Maxima

```
f(x):=x^3+x^2-x+8$ \ sol:allroots(f(x))$
expand(map(f,sol));
```

$$\begin{aligned} & \left[x^3 + x^2 - x + 8 = 4.440892098500626 \right. \\ & \quad \times 10^{-16} i + 1.77635683940025 \times 10^{-15}, \\ & \quad x^3 + x^2 - x + 8 = 1.77635683940025 \times 10^{-15} \\ & \quad \left. - 4.440892098500626 \times 10^{-16} i, x^3 + x^2 - x + 8 = 0.0 \right] \end{aligned}$$



Systems of equations

Maxima

```
load(newton);
newton(x^7-5*x^6+4*x^4-
5*x^2+x+2,0);
```

$-5.763042928902195_B \times 10^{-1}$

Maxima

```
newton(x^7-5*x^6+4*x^4-
5*x^2+x+2,1);
```

$8.194213634964119_B \times 10^{-1}$



Commands Summary

apropos apropos search

linsolve solve linear systems

solve solve algebraic equations

subst substitute values to expressions

allroots numerically estimate all roots of a polynomial

kill “kill(a,b,c);” lets you delete unwanted variables

map map the solutions to a function



Outline

Introduction

Basics

Plotting

Solving Equations

Vectors and Matrices

- Vectors
- Matrices and Linear Algebra

Calculus

- Differential Calculus
- Taylor Series
- Symbolic Integration

Symbolic Manipulations

Vectors



Maxima

```
load(vector);  
functions;
```

[dimension () , type (arg) , coordsystem (sys) , cross (a,
b) , grad (s) , div (v) , curl (a) , laplacian (a) , dotdel (v,
b) , christoffel () , curlgrad (s) , graddiv (v) , divcurl (v)]



Vectors

Maxima

```
v1: [x1, y1, z1]$
v2: [x2, y2, z2]$
v1\ cross\ v2;
```

$$[y_1 z_2 - y_2 z_1, \\ x_2 z_1 - x_1 z_2, x_1 y_2 - x_2 y_1]$$

Maxima

```
v1.v2;
```

$$z_1 z_2 + y_1 y_2 + x_1 x_2$$



Matrices and Linear Algebra

Maxima

```
load(diag);  
A:matrix([a,b,c],[d,e,f],[g,h,i]);  
col(A,1);  
row(A,2);  
invert(A);  
transpose(A);  
determinant(A);  
list_matrix_entries(A);
```

Row major!



Matrices and Linear Algebra

Matrix Products

Maxima

```
load(diag);  
A:matrix([a,b,c],[d,e,f],[g,h,i]);  
B:transpose(matrix([c,d,e]));  
A.B;  
A*A;  
A^2;  
A^^2;  
rank(A);
```



Matrices and Linear Algebra

Matrix properties

Maxima

```
A:matrix([1,2],[4,5]);  
eigenvectors(A);
```

```
[[[3 - 2*sqrt(3), 2*sqrt(3) + 3], [1, 1]],  
 [[1, 1 - sqrt(3)], [1, sqrt(3) + 1]]]
```



Matrices and Linear Algebra

Other functionality

Maxima

```
load(eigen);  
load(linearalgebra);
```



Command Summary

matrix create a matrix

col access column of a matrix

row access row of a matrix

transpose transpose of a matrix

determinant obtain the determinant of a matrix

rank obtain the rank of a matrix

invert invert a matrix

list_matrix_entries list out the matrix in row major

eigenvectors obtain the eigen vectors



Outline

Introduction

Basics

Plotting

Solving Equations

Vectors and Matrices

- Vectors
- Matrices and Linear Algebra

Calculus

- Differential Calculus
- Taylor Series
- Symbolic Integration

Symbolic Manipulations



Differential Calculus

Independant variable specified

Maxima

```
diff(x^n,x,2);
```

$$(n-1) n x^{n-2}$$

Unspecified independant variable

Maxima

```
diff(x^n);
```

$$n x^{n-1} dx + x^n \log x dn$$

Differential Calculus



$$\frac{\partial}{\partial y} \frac{\partial^2}{\partial x^2} (\sin(x) \cos(y - x))$$

Maxima

```
diff(sin(x)*cos(y-x),x,2,y,1);
```

$$2 \sin x \sin (y - x) + 2 \cos x \cos (y - x)$$

Differential Calculus



Maxima

```
declare(a,constant);  
depends([x,y],t);  
diff(x^2+a*y^2,t);
```

$$2ay \left(\frac{dy}{dt} \right) + 2x \left(\frac{dx}{dt} \right)$$

Try out “properties(a);” & “propvars(constant);”



Differential Calculus

Maxima

```
gp:diff(x^(2/3),x);
limit(gp,x,0,plus);
```

$$\frac{2}{3x^{\frac{1}{3}}}$$

∞

Maxima

```
limit(gp,x,0,minus);
limit(gp,x,0);
```

$-\infty$

infinity



Differential Calculus

Maxima

```
dydx(expr,x,y) :=  
diff(expr,x)/diff(expr,y);
```

Maxima

```
dydx(cos(x^2-y^2),x,y);
```

$$\text{dydx}(\text{expr}, x, y) := \frac{\text{diff}(\text{expr}, x)}{\text{diff}(\text{expr}, y)}$$

$$-\frac{x}{y}$$



Taylor (or Laurent) series

Maxima

```
taylor(f(x),x,0,3);
```

$$f(0) + \left(\frac{df(x)}{dx} \Big|_{x=0} \right) x + \frac{\left(\frac{d^2 f(x)}{dx^2} \Big|_{x=0} \right) x^2}{2} + \frac{\left(\frac{d^3 f(x)}{dx^3} \Big|_{x=0} \right) x^3}{6} + \dots$$



Taylor (or Laurent) series

Maxima

```
taylor(cos(x),x,0,10);
```

```
taylor(sin(x),x,0,10);
```

$$1 - \frac{x^2}{2} + \frac{x^4}{24} - \frac{x^6}{720} + \frac{x^8}{40320} - \frac{x^{10}}{3628800} + \dots$$

$$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362880} + \dots$$



Taylor (or Laurent) series

Maxima

```
taylor(cos(x+y), [x,y], [0,0], [2,4]);
factor(ratsimp(\%));
```

$$1 - \frac{y^2 + 2yx + x^2}{2} + \frac{y^4 + 4y^3x + 6y^2x^2 + 4yx^3 + x^4}{24} + \dots$$

$$\frac{24 - 12x^2 + x^4 - 24xy + 4x^3y - 12y^2 + 6x^2y^2 + 4xy^3 + y^4}{24}$$



Taylor (or Laurent) series

Maxima

```
taylor(sin(x+y), [x,0,4], [y,0,4]);
```

$$\begin{aligned} & y - \frac{y^3}{6} + \cdots + \left(\frac{y^4}{24} - \frac{y^2}{2} + 1 + \cdots \right) x + \left(\frac{y^3}{12} - \frac{y}{2} + \cdots \right) x^2 \\ & + \left(-\frac{y^4}{144} + \frac{y^2}{12} - \frac{1}{6} + \cdots \right) x^3 + \left(-\frac{y^3}{144} + \frac{y}{24} + \cdots \right) x^4 + \cdots \end{aligned}$$



Symbolic Integration

Maxima

```
integrate(f(x),x);
```

$$\int f(x) dx$$

Maxima

```
trigsimp(trigreduce
(integrate(sin(x)^3,x)));
```

$$\frac{\cos(3x) - 9 \cos x}{12}$$

Symbolic Integration



Maxima

```
integrate(x^2*exp(-x^2),x);
```

Maxima

```
float(integrate(
x^2*exp(-x^2),x,1,2));
```

$$\frac{\sqrt{\pi} \operatorname{erf}(x)}{4} - \frac{x e^{-x^2}}{2}$$

0.2332527106719843



Symbolic Integration

Maxima

```
expr: 'integrate  
(2*x*(x^2+1)^3,x);
```

$$2 \int x (1 + x^2)^3 dx$$

Maxima

```
exprc: changevar  
(expr,x^2+1-u,u,x);
```

$$\int u^3 du$$



Symbolic Integration

Maxima

```
ev(exprc,nouns);
```

$$\frac{u^4}{4}$$

Maxima

```
subst([u=x^2+1],  
ev(exprc,nouns));
```

$$\frac{(1+x^2)^4}{4}$$



Command Summary

`diff` differentiate expression

`grind` “non-pretty” output

`display2d` set to true for pretty output

`declare` declare a symbol as constant

`depends` establish dependency

`integrate` integrate expression

`changevar` apply a change of variable



Outline

Introduction

Basics

Plotting

Solving Equations

Vectors and Matrices

- Vectors
- Matrices and Linear Algebra

Calculus

- Differential Calculus
- Taylor Series
- Symbolic Integration

Symbolic Manipulations



Simplifying Expressions

factoring and expanding polynomials

Maxima

```
factor(x^2+2*x+1);
```

$$(1 + x)^2$$

Maxima

```
expand((x+1)**2);
```

$$x^2 + 2x + 1$$



Simplifying Expressions

Trigonometric expressions

Maxima

```
trigsimp(sin(x)^2
+cos(x)^2);
```

1

Maxima

```
trigexpand(cos(x+3*y));
```

$\cos x \cos(3y) - \sin x \sin(3y)$



Simplifying Expressions

Rational expressions

Maxima

```
ratsimp((x^2+2*x+1)/
(x+1)+1/(4*x+3));
```

$$\frac{4 + 7x + 4x^2}{4x + 3}$$

Maxima

```
ratexpand((x^2+2*x+1)/
(x+1)+1/(4*x+3));
```

$$\frac{4x^2}{4x + 3} + \frac{7x}{4x + 3} + \frac{4}{4x + 3}$$



Simplifying Expressions

Complex exponentials

Maxima

```
demoivre:true;  
exp(a+b*%i);
```

$$e^a (i \sin b + \cos b)$$

Maxima

```
demoivre:false;  
exp(a+b*%i);
```

$$e^{a+ib}$$



Simplifying Expressions

expand(expr,p,n);

Maxima

```
expand((x+1)^5+(x+1)^(-5)
,3,5);
```

$$\frac{1}{x^5 + 5x^4 + 10x^3 + 10x^2 + 5x + 1 + (1+x)^5}$$

Maxima

```
expand((x+1)^5+(x+1)^(-5)
,5,3);
```

$$\frac{1}{(1+x)^5} + x^5 + 5x^4 + 10x^3 + 10x^2 + 5x + 1$$



Simplifying Expressions

Partial fractions

Maxima

```
partfrac(1/(x^2+3*x+2),x);
```

$$\frac{1}{x+1} - \frac{1}{x+2}$$



Simplifying Expressions

Expressions with logs, exponentials and radicals

Maxima

```
radcan((log(x+x^2)-
log(x))^a/log(1+x)^(a/2));
```

$$(\log(1+x))^{\frac{a}{2}}$$

Maxima

```
radcan((\%e^x-1)/
(1+\%e^(x/2)));
```

$$e^{\frac{x}{2}} - 1$$



Simplifying Expressions

Simplification with rules

Maxima

```
expr:sin(x^2+y^2)-sin(3*x);
eq1:x^2+y^2-x=0;
```

Maxima

```
eq2:sin(x)=y;
scsimp(expr,eq1,eq2);
```

$$\sin(x^2 + y^2) - \sin(3x)$$

$$\sin x = y$$

$$y^2 + x^2 - x = 0$$

$$y - \sin(3x)$$



Command Summary

- `ratsimp` simplify rational expressions
- `ratexpand` expand rational expressions
- `trigsimp` converts all trigonometric quantities in terms of \sin and \cos terms
- `trigreduce` Reduces powers of trigonometric quantities to terms with highest power one
- `trigexpand` expand trigonometric terms in terms of \sin and \cos
- `radcan` simplify expressions with logs, exponentials and radicals



What has NOT been covered

- ▶ Numerical Integration
- ▶ FFT
- ▶ ODEs
- ▶ Advanced plotting
- ▶ Orthonormal series and transforms
- ▶ Inequalities




Setting up imaxima on emacs I

```
(add to list 'load path "/usr/share/maxima/5.37.2/emacs")
(autoload 'maximamode "maxima" "Maxima_mode" t)
(autoload 'imaxima "imaxima"
  "Frontend_for_maxima_with_Image_support" t)
(autoload 'maxima "maxima" "Maxima_interaction" t)
(autoload 'imathmode "imath"
  "Imath_mode_for_math_formula_input" t)
(setq imaxima use maxima mode flag t)
(add to list 'auto mode alist '("\.ma[cx]" .
  maximamode))
(setq imaxima fnt size "Large")

(autoload 'emaximamode "emaxima" "Emaxima" t)
(add hook 'emaximamodehook 'emaxima mark file as emaxima)
(setq emaxima abbreviations allowed t)
```



Setting up imaxima on emacs II





A quick example I

$$F_r = -\mu RL \left(\frac{L}{c} \right)^2 \left[(\omega - 2\dot{\phi}) \frac{\epsilon^2}{(1 - \epsilon^2)^2} + \frac{\pi}{2} \frac{(1 + 2\epsilon^2)\dot{\epsilon}}{(1 - \epsilon^2)^{5/2}} \right]$$

$$F_t = \mu RL \left(\frac{L}{c} \right)^2 \left[(\omega - 2\dot{\phi}) \frac{\pi}{4} \frac{\epsilon}{(1 - \epsilon^2)^{3/2}} + \frac{2\epsilon\dot{\epsilon}}{(1 - \epsilon^2)^2} \right]$$

depends ([e], t)\$

ex:0.3*50.8e 6*cos(%pi/4)\$

ey:0.3*50.8e 6*cos(%pi/4)\$

exd:0.1*ex*omega\$

eyd:0.1*ey*omega\$

vals:[mu=0.00689,h[0]=50.8e 6,L=12.7e 3,R=25.4e 3,
diff(e[x],t)=exd,diff(e[y],t)=eyd,e[x]=ex,e[y]=ey,
omega=10000*2*%pi/60]\$



A quick example II

```
epsilon : sqrt (e[x]**2 + e[y]**2) / h[0] $
phi : atan2 (e[x], e[y]) $
```

```
Fr : mu * R * L * (L / h[0])**2 * ((omega 2 * diff(phi, t)) *
epsilon **2 / (1 epsilon **2)**2 + %pi * (1 + 2 * epsilon **2) *
diff(epsilon, t) / (2 * (1 epsilon **2)**(5/2))) $
```

```
Ft : mu * R * L * (L / h[0])**2 * ((omega 2 * diff(phi, t)) *
% pi * epsilon / (4 * (1 epsilon **2)**(3/2)) +
2 * epsilon * diff(epsilon, t) / (1 epsilon **2)**2) $
```

```
Fx : radcan (Fr * cos(phi) Ft * sin(phi)) $
Fy : radcan (Fr * sin(phi) + Ft * cos(phi)) $
```

```
FxFy : transpose (matrix ([Fx, Fy])) $
```



A quick example III

```
float (subst (vals , FxFy));  
float (subst (vals , diff (FxFy , e[x])));  
float (subst (vals , diff (FxFy , e[y])));  
float (subst (vals , diff (FxFy , diff (e[x] , t))));  
float (subst (vals , diff (FxFy , diff (e[y] , t))));
```

Thank You!



Questions?